

MOVE ALL

MOVE ALL *operand1* **TO** *operand2* [**UNTIL** *operand3*]

Operand	Possible Structure					Possible Formats										Referencing Permitted	Dynamic Definition
Operand1	C	S				A	N				B					yes	no
Operand2		S	A			A										yes	yes
Operand3	C	S					N	P	I							yes	no

Function

The MOVE ALL statement is used to move repeatedly the value of *operand1* to *operand2* until *operand2* is full.

Source Operand - operand1

The source operand contains the value to be moved.

All digits of a numeric operand including leading zeros are moved.

Target Operand - operand2

The target operand is not reset prior to the execution of the MOVE ALL operation. This is of particular importance when using the UNTIL option since data previously in operand2 is retained if not explicitly overlaid during the MOVE ALL operation.

UNTIL Option - operand3

The UNTIL option is used to limit the MOVE ALL operation to a given number of positions in *operand2*. *Operand3* is used to specify the number of positions. The MOVE ALL operation is terminated when this value is reached.

If *operand3* is greater than the length of *operand2*, the MOVE ALL operation is terminated when *operand2* is full.

The UNTIL option may also be used to assign an initial value to a DYNAMIC variable: if *operand2* is a DYNAMIC variable, its length after the MOVE ALL operation will correspond to the value of *operand3*. The current length of a DYNAMIC variable can be ascertained by using the system variable *LENGTH. For general information on DYNAMIC variables, see your Natural User's Guide.

Example

```

/* EXAMPLE 'MOAEX1': MOVE ALL
/*****
DEFINE DATA LOCAL
1 EMPLOY-VIEW VIEW OF EMPLOYEES
  2 PERSONNEL-ID
  2 FIRST-NAME
  2 NAME
  2 CITY
1 VEH-VIEW VIEW OF VEHICLES
  2 PERSONNEL-ID
  2 MAKE
END-DEFINE
/*****
LIMIT 4
RD. READ EMPLOY-VIEW BY NAME
  SUSPEND IDENTICAL SUPPRESS
/*****
FD. FIND VEH-VIEW WITH PERSONNEL-ID = PERSONNEL-ID (RD.)
  IF NO RECORDS FOUND
    MOVE ALL '*' TO FIRST-NAME (RD.)
    MOVE ALL '*' TO CITY (RD.)
    MOVE ALL '*' TO MAKE (FD.)
  END-NOREC
/*****
  DISPLAY NOTITLE (ES=OFF IS=ON ZP=ON AL=15)
    NAME (RD.) FIRST-NAME (RD.) CITY (RD.)
    MAKE (FD.) (IS=OFF)
/*****
  END-FIND
END-READ
END

```

NAME	FIRST-NAME	CITY	MAKE
ABELLAN	*****	*****	*****
ACHIESON	ROBERT	DERBY	FORD
ADAM	*****	*****	*****
ADKINSON	TIMMIE	BEDFORD	GENERAL MOTORS

Equivalent reporting-mode example: See the program MOAEX1R in the library SYSEXRM.